
ABSTRACT 

The Collaborative Longitudinal Evaluation of Keratoconus (CLEK) Study was a multi-center observational study 
funded by the National Eye Institute to characterize vision, corneal changes, and quality of life in keratoconus 
patients, and to determine the progression of the disease over time.  A total of 1,209 keratoconus patients at 16 
participating clinics were examined annually for 8 years. 

Because CLEK is the first systematic, longitudinal study of keratoconus, an eye condition about which very little is 
known, study data will be an especially valuable resource to investigators outside the study. 

To provide access to the data following the study’s completion, we have produced a web-based archive containing 
extensive documentation of the study and its database.  The archive includes traditional text and standard SAS 
documentation.  We supplemented the traditional documentation with PDF images of all study data collection forms, 
annotated with SAS dataset variable names for the associated form fields and with links from each variable name to 
details about that variable and a summary of its contents. 

We will describe the methods and tools used to produce this archive, and illustrate with both code fragments and 
sample screens. 
 

BACKGROUND 

The database for the CLEK Study is composed of 21 SAS datasets: one for each of the four major data collection 
forms used in the study, several administrative datasets, and datasets built from raw data sent from our two reading 
centers.  More than 1000 variables are represented. 

Since the study began enrollment in 1994, numerous revisions have been made to the study protocol and forms and, 
consequently, numerous additions and deletions of variables from the associated SAS datasets.  A common format 
library is shared by all study datasets, and formats in the library have also undergone changes over the years. 

Data collection for the study ended in 2004, with the last patient follow-up visit completed in August of that year.  
Analysis, preparation of manuscripts, and the closeout and archiving of study materials have continued since then. 

To archive CLEK study materials and data, we have produced an archive website containing historical information 
and material useful for those interested in understanding the study and interpreting its results and publications.   
 

DESIGN CONCEPT 

Three objectives went into the design and selection of content for this Web Archive: 

 To provide a history of the study and overview of its organization and operation. 

 To provide sufficient details of the protocol, procedures and data collection forms to aid in interpreting the 
study’s published results 

 To provide an introduction to the compiled database, its organization and contents, for those outside 
investigators who may wish to use it for further analysis. 

The archive includes standard text documents, including the study’s Manual of Procedures, a historical overview and 
timeline, bibliography of published manuscripts and publications, and more.  For this SUGI presentation, however, we 
focus on the documentation of the study’s SAS database, including both traditional SAS documentation and the 
extended documentation developed specifically for this archive. 

 

TRADITIONAL DOCUMENTATION OF THE SAS DATABASE 

PROC CONTENTS and PROC FORMAT 

Traditional documentation of a SAS database begins with 
printing of the output from PROC CONTENTS for all datasets in 
the library.  For our archive, we used the Output Delivery System 
(ODS) to format the output as a Portable Document Format 
(PDF) file, and include both a directory of all the datasets (not 
shown) and contents for each individual dataset (see Figure 1: 
PROC CONTENTS). 

The PDF output file for this and all of our archive documents 
described below, is routed to a directory on our web server, 
where it can be linked from the archive menu pages. 

 

 

 

 

 

We also used ODS to produce PDF output from PROC FORMAT, 
documenting the many custom formats shared by variables in our 
database (see Figure 2: PROC FORMAT). 

 

 

 

 

 

 

 

 

 

Figure 1: PROC CONTENTS 

Figure 2: PROC FORMAT 

CODEBOOK 

Throughout our study we have used a Codebook, 
combining in one concise document the information from 
PROC CONTENTS and PROC FORMAT for all variables 
in all datasets used by the study.  The codebook provides 
a more comprehensive view of the database than 
CONTENTS or FORMAT viewed separately, and is more 
intuitive and accessible to non-programmers. 

The codebook is produced, entirely in SAS, by merging 
the output from CONTENTS and FORMAT within a 
datastep that formats the information for easy, convenient 
access by both SAS users and those unfamiliar with 
SAS.  (Note: this program was written in 1995 and has 
been in use continuously since then, without modification 
to use features since added in more recent versions of 
SAS.) 

At the study’s conclusion, we produced a PDF version of 
the codebook for inclusion in the archive web site 
(illustrated in Figure 3). 

 

 

 

 

 

 

 

BEYOND TRADITIONAL: ANNOTATED FORMS  

In order to facilitate an in-depth understanding of the relationship between the data collection forms used in the study 
and the elements of the database produced from those forms, we built an annotated copy of each form with “hot links” 
providing details of the variables associated with each element on the form.  Building each annotated form required 
two steps: generating a PDF file containing the detailed information about each variable, and creating an image of the 
form using the associated variable name to label the field and provide a clickable link to the details. 

USE SAS MACROS TO PRODUCE PDF FILES WITH DETAILS ABOUT EACH VARIABLE 

We designed a simple set of SAS macros to combine standard variable descriptors (variable name, label, type, 
length, format, informat) for each field with a statistical summary of the contents of that field, using PROC MEANS for 
numeric fields, and PROC FREQ for character fields.    

(Note: The complete macro code and a typical invocation are included as an appendix to this paper;  in the discussion 
which follows, only fragments are included.) 

The first macro, %ANNOTATE, is invoked once for each dataset to be processed.  It receives five parameters: the 
library and dataset names (LIB, DSN), an optional dataset alias (DSNREF), the name of a variable in the dataset to 
be used for statistical sub-grouping (STATSBY), and a list of variables for which a statistical summary is not to be 
produced (EXCL).  For each variable in the dataset, %ANNNOTATE first gathers dictionary information about the 
variable using SAS File I/O functions VARNAME, VARLABEL, VARLEN, VARTYPE, VARFMT, VARINFMT. 
 
   %let dsid=%sysfunc(open(&dset));  
   %if &dsid %then %do;   
       %let nfmts=0; %let ninfmts=0;  *--initialize counters--;  
       %let nvars=%sysfunc(attrn(&dsid,nvars)); 
       %do i=1 %to &nvars;   

Figure 3: Codebook 

           %let existfmt=0;  %let existinf=0;   
           %let vname =%sysfunc(varname (&dsid,&i));   
           %let vlabel=%qsysfunc(varlabel(&dsid,&i));  
           %let vlabel=%sysfunc(compress(&vlabel,%str(%'))); 
           %let vlen  =%sysfunc(varlen  (&dsid,&i));   
           %let vtype =%sysfunc(vartype (&dsid,&i));   
           %let vfmt  =%sysfunc(varfmt  (&dsid,&i));   
             %if x&vfmt=x %then %let vfmt=Undefined;   
             %else %do;   
                *-- first remove $ and periods from the format--; 
                %if %substr(&vfmt,1,1)=$ %then %let vfmt=%substr(&vfmt,2);   
                %let vfmt=%substr(&vfmt,1,%eval(%length(&vfmt)-1));  
                *--determine if format/informat is in the library--; 
                %if &vtype=C %then  
                    %let existfmt=%sysfunc(exist(library.formats.&vfmt..formatc)); 
                %if &vtype=N %then  
                    %let existfmt=%sysfunc(exist(library.formats.&vfmt..format));  
                *--run MAKEFMT macro to build PDF file for this format--;   
                %if &existfmt %then %makefmt(&vfmt,&vtype,format);  
                %end; 

%ANNOTATE then formats and includes that information in a set of centered TITLE statements.   

 
   *--Use TITLE statements to add dataset & var details to freq/MEANS output--;    
   title1 j=c "**********************************************"; 
   title2 j=c "Dataset is &DSET";   
   title3 j=c "Variable Name is &VNAME"  ; 
   title4 j=c "Variable Label is '&VLABEL'";   
   title5 j=c "Variable Type is &VTYPE";   
   *--TITLE statements for formats/informats include imbedded URLs--;  
   %let click=Click here for details;   
   %if (&existfmt) %then  
    title6 j=c "^S={URL='format_&VFMT..PDF'} Variable Format is &vfmt - &Click" ;; 
   %if (&existfmt=0)%then  
    title6 j=c "Variable Format is &vfmt";; 
   %if (&existinf) %then  
    title7 j=c "^S={URL='infmt_&VINFMT..PDF'} Variable Informat is &vinfmt--&Click";;   
   %if (&existinf=0) %then  
    title7 j=c "Variable Informat is &vinfmt";; 
   title8 j=c "**********************************************"; 

Finally,  %ANNOTATE runs either the MEANS procedure (for numeric variables) or the FREQ procedure (for 
character variables).  PROC FREQ is run as a two-way crosstabulation and MEANS is run with a CLASS statement, 
using the specified STATSBY variable to group the data into classification levels. The output of FREQ or MEANS, 
with the added information in the TITLE  statements, is formatted as PDF and routed to a file in the web archive 
directory. 
 
   *--Summarize by value of specified STATSBY variable, usually by year--;   
   *--Omit the summary if var is on excluded list*/ 
   %if %index(x&excl,&vname)=0  %then %do;   
      %if &vtype=C  %then %do; 
         proc freq data=&dset;    
           tables &STATSBY*&vname;    
           %if &statsby=EXAMDATE %then FORMAT &STATSBY YEAR4.; 
           ;run;   
           %end;  
       %if &vtype=N %then %do;  
          proc means data=&dset; var &vname;  
            %if &statsby=EXAMDATE %then FORMAT &STATSBY YEAR4.;  
            %if &vfmt=:MMDDYY %then FORMAT &vname YEAR4.;  
            ;CLASS &STATSBY;   
            ;run;   
          %end;  
       %end;  
    %else %do;    

       data _null_; file print footnotes;  
       put "No Statistical Summary for this variable"; run; 
       %end;  

%ANNOTATE also invokes the second macro, %MAKEFMT, to gather details of the custom format and/or informat, if 
any, for that variable, and produces a PDF page containing that information.  The link to the format/informat detail 
page is imbedded in the associated TITLE statement produced by %ANNOTATE (above).  

 
%MACRO MAKEFMT(vfmt,vtype,ftype);    
   title; footnote; 
   %let file=&basefile.\&ftype._&vfmt..PDF ;    
   %let existfile=%sysfunc(fileexist(&file));   
   %if &existfmt and not &existfile %then %do;  
      ODS PDF file="&file" style=jfastyle bookmarklist=hide; 
      proc format fmtlib lib=library;   
        %if &vtype=C %then select $&vfmt.; %else select  &vfmt.;; run;  
      %end; 
   ODS PDF close;   
   %MEND MAKEFMT;   

The result after running %ANNOTATE for each dataset is a PDF file for each variable incorporating in one screen per 
variable detailed dictionary information for the SAS variable, links to the associated format and/or informat, and a brief 
bivariate summary of the data in the CLEK database.  The next step in building our archive was to provide access to 
these detail pages from copies of the original CLEK data collection forms. 

USE ADOBE ACROBAT TO PREPARE 
ANNOTATED FORMS 

We used Adobe Acrobat Professional 
7.0 to convert each of the forms used 
in the study from the original Microsoft 
Word document to a PDF file. We then 
added text form field annotation to 
display the associated SAS variable 
name on or adjacent to the entry field 
or checkbox for each variable.   

A sample from one of our annotated 
forms is illustrated in Figure 4, at right.  
The grey boxes are the annotations 
containing variable names. 

 

 

 

 

 

 

 

 

 

Figure 4: Annotated Form 

We then assigned a mouse up function to each variable name, linking the variable name to the PDF file that displays 
the variable detail page (PDF) created by %ANNOTATE. The mouse up function opens the PDF file whenever the 
user clicks and releases the mouse button over the variable name.  Figure 5 illustrates how, by clicking on the 
variable name CONSENT in the annotated form, the user can then view the variable detail page for that variable.  The 
second arrow in the figure illustrates how the user can then click on the format/informat name (if any) to get details of 
the formatting of the variable’s values. 

 

 

 

 

 

 

 

 

 

 

 

 

 

 

 

 

 

 

 

 

CONCLUSION 

We believe the CLEK archive, and the tools used to build it, can serve as a model for documentation of any research 
study or business application, not only as an archive at the conclusion of the project, but as a readily accessible, 
easily useable reference while the project is on-going or under development.  The tools we developed are adaptable 
to any SAS database and easily customized to meet any special circumstances.   

We invite you to visit our web site at https://vrcc.wustl.edu/clekarchive to see the full scope of documentation included 
in our archive and to try out the interactive annotated forms for yourself. 

Figure 5: Linking to Variable Details 

HIPAA COMPLIANCE NOTE 

The CLEK Study carefully guards the confidentiality of its participants and is in full compliance with the Health 
Insurance Portability and Accountability Act (HIPAA).  The web archive does not include access to any raw study 
data, and no patient identifiers are included in the summary statistics displayed on the annotated form detail pages.  
All data made available to researchers, independently of the web site, is carefully cleansed of patient identifying 
information in full compliance with HIPAA prior to release. 

APPENDIX: FULL TEXT OF ANNOTATE AND MAKEFMT MACROS, AND TYPICAL INVOCATION 
 
*-----------------------------------------------------------------------;   
options nodate nonumber formchar='|----|+|---+=|-/\<>*';   
%get_time; *--formats current date for use in titles/footnotes--;   
footnote "Program: &progname   Prepared: &curdate";    
libname MASTER "S:\MASTER";  
%let basefile=        W:\CLEKARCHIVE\Website\AnnotatedForms; 
%let basehref=file:///W:/CLEKARCHIVE/Website/AnnotatedForms; 
ods listing close;  
*-----------------------------------------------------------------------;   
*--CREATE MACRO ANNOTATE                                              --;   
*-----------------------------------------------------------------------;   
%MACRO ANNOTATE(LIB,DSN,DSNREF,STATSBY,EXCL);   
%global nfmts ninfmts;  %global existfmt existinf;  
%let dset=&lib..&dsn;  
%let nchar=0; %let nnum=0;    
%let excl=%upcase(&excl); 
%let statsby=%sysfunc(upcase(&statsby)); 
%let excl   =%sysfunc(upcase(&excl));    
ods results off; ods trace off;   
ods pdf file="&basefile.\&dsnREF._contents.pdf"  
       style=styles.jfastyle bookmarklist=hide;   
ods pdf title=" "; 
ODS pdf exclude contents.dataset.enginehost; 
ods noproctitle; 
ods proclabel='';  
title2;  
proc contents data=&dset; run;    
ODS PDF close;   
%let dsid=%sysfunc(open(&dset));  
%if &dsid %then %do;   
    %let nfmts=0; %let ninfmts=0;  *--initialize counters--;  
    %let nvars=%sysfunc(attrn(&dsid,nvars)); 
    %do i=1 %to &nvars;   
        %let existfmt=0;  %let existinf=0;   
        %let vname =%sysfunc(varname (&dsid,&i));   
        %let vlabel=%qsysfunc(varlabel(&dsid,&i));  
        %let vlabel=%sysfunc(compress(&vlabel,%str(%'))); 
        %let vlen  =%sysfunc(varlen  (&dsid,&i));   
        %let vtype =%sysfunc(vartype (&dsid,&i));   
        %let vfmt  =%sysfunc(varfmt  (&dsid,&i));   
          %if x&vfmt=x %then %let vfmt=Undefined;   
          %else %do;   
             *-- first remove $ and periods from the format--; 
             %if %substr(&vfmt,1,1)=$ %then %let vfmt=%substr(&vfmt,2);   
             %let vfmt=%substr(&vfmt,1,%eval(%length(&vfmt)-1));  
             *--determine if format/informat is in the library--; 
             %if &vtype=C %then  
                 %let existfmt=%sysfunc(exist(library.formats.&vfmt..formatc)); 
             %if &vtype=N %then  
                 %let existfmt=%sysfunc(exist(library.formats.&vfmt..format));  
             *--run MAKEFMT macro to build PDF file for this format--;   
             %if &existfmt %then %makefmt(&vfmt,&vtype,format);  
             %end; 
        %let vinfmt=%sysfunc(varinfmt(&dsid,&i));   
          %if x&vinfmt=x %then %let vinfmt=Undefined;   
          %else %do;   
             %if %substr(&vinfmt,1,1)=$ %then %let vinfmt=%substr(&vinfmt,2);   
             %let vinfmt=%substr(&vinfmt,1,%eval(%length(&vinfmt)-1)); 
             *--determine if format/informat is in the library--; 
             %if &vtype=C %then  
                 %let existinf=%sysfunc(exist(library.formats.&vinfmt..infmtc));  
             %if &vtype=N %then  
                 %let existinf=%sysfunc(exist(library.formats.&vinfmt..infmt)); 
             *--run MAKEFMT macro to build PDF file for this input format--;    
             %if &existinf %then %makefmt(&vinfmt,&vtype,infmt); 

             %end; 
        ODS PDF file="&basefile.\&dsnREF._&vname..PDF"  
                style=jfastyle bookmarklist=hide startpage=never;    
        ods escapechar="^";   
        *--Use TITLES to add dataset and var details to freq/MEANS output--;    
        title1 j=c "**********************************************"; 
        title2 j=c "Dataset is &DSET";   
        title3 j=c "Variable Name is &VNAME"  ; 
        title4 j=c "Variable Label is '&VLABEL'";   
        title5 j=c "Variable Type is &VTYPE";   
        *--TITLE statements for formats/informats include imbedded URLs--;  
        %let click=Click here for details;   
        %if (&existfmt) %then  
          title6 j=c "^S={URL='format_&VFMT..PDF'} Variable Format is &vfmt-&Click" ;; 
        %if (&existfmt=0) %then title6 j=c "Variable Format is &vfmt";; 
        %if (&existinf) %then title7 j=c  
            "^S={URL='infmt_&VINFMT..PDF'} Variable Informat is &vinfmt - &Click" ;;   
        %if (&existinf=0) %then title7 j=c "Variable Informat is &vinfmt";; 
        title8 j=c "**********************************************"; 
        *--Summarize by value of specified STATSBY variable, usually by year--;   
        *--only do Stats summary if not excluded and if not a certification field*/ 
        %if %index(x&excl,&vname)=0 and %sysfunc(index(&vname,CERT))=0 %then %do;   
           %if &vtype=C  %then %do; 
              proc freq data=&dset;    
                tables &STATSBY*&vname;    
                %if &statsby=EXAMDATE or &statsby=DOD %then FORMAT &STATSBY YEAR4.;  
                  ;run;   
                %end;  
            %if &vtype=N %then %do;  
               proc means data=&dset; var &vname;  
                 %if &statsby=EXAMDATE or &statsby=DOD  
                     %then FORMAT &STATSBY YEAR4.;  
                  %if &vfmt=:MMDDYY %then FORMAT &vname YEAR4.;  
                  ;CLASS &STATSBY;   
                  ;run;   
                %end;  
            %end;  
         %else %do;    
            data _null_; file print footnotes;  
            put "No Statistical Summary for this variable"; run; 
            %end;  
        ODS PDF close;  *--stop ODS output--;   
        %end;    
    %let rc=%sysfunc(close(&dsid));  
    %end;    
%ELSE %PUT Attempt to open &dset failed - %sysfunc(sysmsg()); 
%MEND ANNOTATE;  
 
*----------------------------------------------------------------------------; 
*--CREATE MACRO MAKEFMT(VFMT,VTYPE,FTYPE) - FTYPE is either FORMAT or INFMT--; 
*----------------------------------------------------------------------------; 
%MACRO MAKEFMT(vfmt,vtype,ftype);    
title; footnote; 
%let file=&basefile.\&ftype._&vfmt..PDF ;    
%let existfile=%sysfunc(fileexist(&file));   
%if &existfmt and not &existfile %then %do;  
   ODS PDF file="&file" style=jfastyle bookmarklist=hide; 
   proc format fmtlib lib=library;   
     %if &vtype=C %then select $&vfmt.; %else select  &vfmt.; 
     ; run;  
   %end; 
ODS PDF close;   
%MEND MAKEFMT;   

*----------------------------------------------------------------------------; 
*-- TO RUN THE ANNOTATE MACRO FOR EACH SELECTED DATASET                    --; 
*-- PARAMETERS ARE LIBRARY, DATASET, STATSBY, and list ofEXCLUDED Variables -; 
*-- EXCLUDED VARS DO NOT GET A PROC FREQ OR MEANS                     --; 
*----------------------------------------------------------------------------; 
%annotate(master,DT,DT,dod     ,idnum namecode formdate editdate movedate dod);    
%annotate(master,EL,EL,examdate,idnum namecode examdate editdate movedate consdate);   
*--etc, for each dataset--;  

